

Date: Fri, 18 Mar 94 04:30:29 PST
From: Ham-Homebrew Mailing List and Newsgroup <ham-homebrew@ucsd.edu>
Errors-To: Ham-Homebrew-Errors@UCSD.Edu
Reply-To: Ham-Homebrew@UCSD.Edu
Precedence: Bulk
Subject: Ham-Homebrew Digest V94 #65
To: Ham-Homebrew

Ham-Homebrew Digest Fri, 18 Mar 94 Volume 94 : Issue 65

Today's Topics:

 Best cars for mobile HF/VHF??
 Crystal Filter Design program & details
 Meter Shunts, etc
 Plans for DCD-3105 board

Send Replies or notes for publication to: <Ham-Homebrew@UCSD.Edu>
Send subscription requests to: <Ham-Homebrew-REQUEST@UCSD.Edu>
Problems you can't solve otherwise to brian@ucsd.edu.

Archives of past issues of the Ham-Homebrew Digest are available
(by FTP only) from UCSD.Edu in directory "mailarchives/ham-homebrew".

We trust that readers are intelligent enough to realize that all text
herein consists of personal comments and does not represent the official
policies or positions of any party. Your mileage may vary. So there.

Date: Thu, 17 Mar 94 10:59:57 GMT
From: ihnp4.ucsd.edu!agate!doc.ic.ac.uk!lyra.csx.cam.ac.uk!pavo.csi.cam.ac.uk!
pipex!uknet!uos-ee!ee.surrey.ac.uk!M.Willis@network.ucsd.edu
Subject: Best cars for mobile HF/VHF??
To: ham-homebrew@ucsd.edu

In article <CMIACE.D9C@hpbmqmoca.sqf.hp.com>, dstock@hpbmqmoca.sqf.hp.com (David
Stockton) writes:

|>
|> I'm happy with my choice, a Diesel powered Range-Rover derivative
|> called a "Discovery"
|>
|> Give serious thought to Diesels, no ignition, no computers
|>
|>
|> David GM4ZNX

Yes, but at a mere 18,000 pounds not many can afford such a car. Practically, I
found the Cavallier reasonable RF quiet. Fiat Uno, too noisy. Diesels are
definatly better, they have a bigger battery too.

Mike

Date: Thu, 17 Mar 1994 01:05:04 GMT
From: ihnp4.ucsd.edu!usc!howland.reston.ans.net!usenet.ins.cwru.edu!nshore!
seastar!jjw@network.ucsd.edu
Subject: Crystal Filter Design program & details
To: ham-homebrew@ucsd.edu

For the past week or two I have been figuring out how to design crystal IF filters using the technique described in the May 1982 article in QST by Wes Hayward. Rather than duplicate his circuit, I used my TW-1 DDS as the signal generator and read the results with an oscscope.

I matched a large pile of surplus 10MHz computer crystals cannibalized from dead boards, and found (from about 60 crystals) 29 crystals that matched within 10Hz (my limit of accuracy in measurement), and they also matched for 3db bandwidth, parallel capacitance (as measured by my radio shack multimeter) and resistance.

Armed with this data, I wrote a simple BASIC program to perform the calculations discussed in the article. As Wes also mentioned the importance of re-resonating each section, I added in the calculations for the series capacitors for each crystal to get them all to resonate more-or-less at the same frequency.

I built a 5 pole Chebychev filter with 2.2kHz bandwidth, and hooked it up (wrongly) to my analyzer (I didn't have the right connector and the antenna in isn't 50 ohms). I got mediocre results, due to improper termination. Tonight, I found the right connector and re-ran the measurements.

With no tuning at all, using the closest stock capacitance to the calculated values, I wound up with a 2.1kHz 3db BW. Ripple was designed to be 0.1db, but I measured about 1db of ripple. The insertion loss was measured to be 2.3db. The 60:6 shap factor is 3.34. It drops sharply on the low side, and is more rounded on the high side.

I'd say it was a success, all in all. As I've had quite a few requests for the program to calculate this, here it is. Have fun with it, and I hope to be on the air with my own SSB unit some day soon.

----- cut here for program -----

```
10 REM xfilter.bas  by n9jzw  jjw
20 REM absolutely no rights reserved - genuine free public domain software
30 REM calculates 2 through 8 pole 0.1db ripple chebychev or butterworth filters
40 REM for how to measure the parameters this program needs, see May 1982 QST's
50 REM article by Wes Hayward
60 REM and/or a future article by n9jzw on the same subject
```

```

70 CLS
80 PRINT "N-pole Crystal Filter designer"
90 DIM Q[10,2],TABLE[10,10,2],C[10],CADD[10],RF[10]
100 GOSUB 790
110 INPUT "Enter 1 for Chebychev 0.1db ripple, or 2 for Butterworth";TYPE
120 INPUT "Crystal 3db bandwidth in Hz";DELTA F
130 INPUT "Crystal series resonant frequency in MHz";F0
140 INPUT "Crystal resistance in ohms";R0
150 INPUT "Crystal parallel capacitance (usually near 5)";CP
160 INPUT "Filter bandwidth in Hz";B
170 INPUT "Order of filter";ORD
180 IF ORD >1 AND ORD <9 THEN 210
190 PRINT "Must be between 2 and 8"
200 GOTO 170
210 FOR X = 1 TO ORD-1
220 C[X] = INT(1326 * (DELTA F/(B*TABLE[ORD,X,TYPE]*F0)) - 2*CP + .5)
230 NEXT X
240 LM = 19.1/DELTA F
250 CM = 1.326E-15 * DELTA F/(F0*F0)
260 REND = (120*B)/(Q[ORD,TYPE]*DELTA F) - R0
270 PRINT "Rend =";REND
280 INPUT "Enter terminating resistance ";R
290 IF R > REND THEN 320
300 PRINT "Terminating resistance must be larger than";REND
310 GOTO 280
320 CEND = INT(((159000!/(R*F0)) * SQR(R/REND-1) - CP + .5)
330 W = 2*3.14159*F0*1000000!
340 CS = (1/(R0*R0)+W*W*CEND*CEND*1E-12*1E-12)/(W*W*CEND*1E-12)
350 C = 1/(1/CS+1/CM+1/(1E-12*C[1]))
360 FEND = SQR(1/(4*3.14159*3.14159*LM*C))
370 RF[1] = FEND
380 RF[ORD] = FEND
390 FOR X = 1 TO ORD-2
400 C = 1/(1/(1E-12*C[X])+1/(1E-12*C[X+1])+1/CM)
410 F = SQR(1/(4*3.14159*3.14159*LM*C))
420 RF[X+1] = F
430 NEXT X
440 Y = Y1 = 0
450 FOR X = 1 TO ORD
460 IF RF[X] < Y THEN 490
470 Y = RF[X]
480 Y1 = X
490 NEXT X
500 CLS
510 IF TYPE = 1 THEN 540
520 PRINT "Butterworth Filter, ";
530 GOTO 550
540 PRINT "Chebychev Filter, ";

```

```

550 PRINT "BW =";B;"Hz at";FO;"MHz,";ORD;"pole with";R;"ohm terminations."
560 FOR X = 1 TO ORD
570 XL = 2 * 3.14159 * Y * LM
580 RSLT = 1 / (XL * 2 * 3.14159 * Y)
590 IF X = 1 OR X = ORD THEN 620
600 Z = 1/RSLT-1/CM-1/(C[X]*1E-12)-1/(C[X-1]*1E-12)
610 GOTO 630
620 Z = 1/RSLT-1/CM-1/(CS)-1/(C[1]*1E-12)
630 CADD[X] = INT(1/Z * 1E+12+.5)
640 NEXT X
650 PRINT "Cend =",CEND
660 FOR X = 2 TO ORD
670 PRINT USING "C # # =" ;X-1,X;
680 PRINT ,C[X-1]
690 NEXT X
700 PRINT "Cend =",CEND
710 PRINT "Series caps"
720 FOR X = 1 TO ORD
730 IF RF[X] = Y THEN 760
740 PRINT "X";X;"=",CADD[X]
750 GOTO 770
760 PRINT "X";X;"=",0
770 NEXT X
780 GOTO 1080
790 FOR Z = 1 TO 2
800 X = 1
810 READ Q[X,Z]
820 IF Q[X,Z] = 0 THEN 880
830 FOR Y = 1 TO 10
840 READ TABLE[X,Y,Z]
850 NEXT Y
860 X = X + 1
870 GOTO 810
880 NEXT Z
890 RETURN
900 DATA 1,1,0,0,0,0,0,0,0,0,0
910 DATA 1.6382,.7106,0,0,0,0,0,0,0,0,0
920 DATA 1.4328,.6618,.6618,0,0,0,0,0,0,0,0
930 DATA 1.3451,.685,.5421,.685,0,0,0,0,0,0,0
940 DATA 1.3013,.7028,.5355,.5355,.7028,0,0,0,0,0,0
950 DATA 1.277,.715,.539,.518,.539,.715,0,0,0,0,0
960 DATA 1.262,.722,.542,.516,.516,.542,.722,0,0,0,0
970 DATA 1.25,.727,.545,.516,.510,.516,.545,.727,0,0,0
980 DATA 0
990 DATA 1,1,0,0,0,0,0,0,0,0,0
1000 DATA 1.414,.7071,0,0,0,0,0,0,0,0,0
1010 DATA 1,.7071,.7071,0,0,0,0,0,0,0,0
1020 DATA .7654,.8409,.4512,.8409,0,0,0,0,0,0,0

```

```
1030 DATA .6180,1,.5559,.5559,1,0,0,0,0,0,0
1040 DATA .518,1.169,.605,.518,.605,1.169,0,0,0,0,0
1050 DATA .445,1.342,.667,.527,.527,.667,1.342,0,0,0,0
1060 DATA .391,1.52,.734,.551,.510,.551,.734,1.52,0,0,0
1070 DATA 0
1080 END
```

----- cut here -----

--

```
While (its_not_working())
    mess_with_it();
```

John Welch, N9JZW
jjw@seastar.org

Date: Thu, 17 Mar 1994 03:45:13 GMT
From: ihnp4.ucsd.edu!library.ucla.edu!europa.eng.gtefsd.com!emory!wa4mei!ke4zv!
gary@network.ucsd.edu
Subject: Meter Shunts, etc
To: ham-homebrew@ucsd.edu

In article <1994Mar14.162444.11831@ccd.harris.com> drs@ccd.harris.com (Doug Snowden) writes:

>I have a few meters laying around, and I need to change the fullscale range
>of most of them, to me useful. For example, I have a meter that has an
>apparent fullscale range of 750 ma. I would like to change this one to a
>1.2 amp fullscale. My question is: Are there any ingenious methods out there
>to modify the range? Like some sort of special material that has a known
>resistance? I know I can figure out the resistance of a foot of # xx wire
>for a shunt. Also, I haven't found any sort of chart that has the resistance
>of wire. What determines the internal resistance of a meter? The type of
>movement?

You increase the full scale reading of a current meter by putting a shunt across it so that it only gets a portion of the current through it. The current divides in inverse proportion to the resistance values, IE with a 10 ohm resistor shunted by a 1 ohm resistor the 1 ohm resistor will receive 10 times as much current as the 10 ohm resistor. A good resistance material for the shunt is nichrome wire, the stuff used as the heating element in toasters. For a small value resistance, a simple length of copper wire works. The ARRL Handbook has a copper wire table listing the resistance of various gauges per 1000 feet.

A quick and easy method to find the internal resistance of an unknown meter is to put a pot and power supply in series with it and adjust the pot for a full scale reading. Now put another pot across the meter and adjust it until the meter reads half scale. Read the value of the pot with an ohmmeter (out of circuit) and you'll have the internal resistance of the unknown meter.

The internal resistance of a meter is a function of meter type and meter sensitivity (obviously). In a moving coil meter, you can think of the movement as a motor stalled against a spring, and the resistance is a result of the motor stall torque against the spring. In a hot wire ammeter, the resistance is a real resistor that heats a bimetallic strip that drives the meter pointer.

Gary

--

Gary Coffman KE4ZV		You make it,		gatech!wa4mei!ke4zv!gary
Destructive Testing Systems		we break it.		uunet!rsiatl!ke4zv!gary
534 Shannon Way		Guaranteed!		emory!kd4nc!ke4zv!gary
Lawrenceville, GA 30244				

Date: Wed, 16 Mar 1994 22:18:44 GMT
From: ihnp4.ucsd.edu!sdd.hp.com!portal!combdyn!lawrence@network.ucsd.edu
Subject: Plans for DCD-3105 board
To: ham-homebrew@ucsd.edu

I have some TNCs and some baycom boards on my multiport Packet BBS.

I was reading through the manual for one of the TNCs (DRSI DPK-2), and it talks about a DCD board option that basically goes inbetween the TCM3105 chip and its socket.

Having DCD has shown itself to be nice with the Software DCD that the KPC-3 TNC does. I have problems with Intermod and frontend overload problems, etc.

Without having to buy the board itself, are there plans on how to build this DCD unit?

And, would anybody know if it would work on Baycom modems built around the same chip.

--

WORK: lawrence@combdyn.com		PHONE 403 529 2162		FAX 529 2516		VE6LKC
HOME: dreamer@lhaven.uumh.ab.ca		403 526 6019		529 5102		VE6PAQ

Praxis BBS - 529 1610 | CYSNET BBS - 526 4304 | Lunatic Haven BBS - 526 6957

disclamer = (working_for && !representing) + (Combustion Dynamics Ltd.);

End of Ham-Homebrew Digest V94 #65
